
179. Trečiasis tūkstantmetis. Pastebėkime, kad likusių dienų ir likusių valandų, minučių, sekundžių skaičių galime ieškoti visiškai nepriklausomai.


Pirmiausia suskaičiuojame kiek dienų turi kiekvienas 2000–ųjų metų mėnuo: 


function mėnesio_dienos (mėnuo: word): word;


{ kiek duotasis 2000 metų mėnuo turi dienų }


begin


  case mėnuo of


    4, 6, 9, 11: mėnesio_dienos := 30;


    2:   mėnesio_dienos := 29; { 2000-ieji – keliamieji metai }


    else mėnesio_dienos := 31;


  end;


end; { mėnesio dienos }





Tuomet nebesunku rasti kiek dienų liko iki tūkstantmečio pradžios:


procedure rasti_dienas (var dienos: word);


{ suskaičiuoja kiek dienų liko iki tūkstantmečio pradžios }


  var i, metai, mėnuo, diena, sav_diena: word;


begin


  getdate (metai, mėnuo, diena, sav_diena);


  { imame likusias dabartinio mėnesio dienas }


  dienos := mėnesio_dienos(mėnuo) - diena;


  for i := mėnuo+1 to 12 do


    dienos := dienos + mėnesio_dienos (i)


end; { rasti_dienas }





Beliko apskaičiuoti valandų, minučių ir sekundžių skaičių:


procedure rasti_val_min_sek (var valandos, minutės, 


                                 sekundės: word);


{ suskaičiuoja valandų, minučių bei sekundžių, likusių iki tūkstantmečio }


{ pradžios, skaičių }


  var val, min, sek, sek100: word;


begin


  gettime (val, min, sek, sek100);


  valandos := 23 - val;


  minutės := 59 - min;


  sekundės := 59 - sek;


end; { rasti val_min_sek }





Norint patikrinti, ar teisingai veikia ši programa galima keisti kompiuterio laikrodžio parodymus ir bandyti ją su įvairiais duomenimis. Paleidę programą 2000 m. sausio 1 d. 0 val., 0 min. 1 sek. gavome rezultatą:


365 23 59 58








180. Slaptaraštis. Šifruojant raktas reikalingas tik tam, kad žinotume kokia tvarka reikia imti stulpelius. Tad pirmasis darbas – sudaryti lentelę iš kurios būtų aišku kokia tvarka reikia imti stulpelius. Pavyzdžiui, duotas raktas septyni. Jo raidės sunumeruojamos taip:


�
1�
2�
3�
4�
5�
6�
7�
�
Raidė�
s�
e�
p�
t�
y�
n�
i�
�
Jos numeris�
5�
1�
4�
6�
7�
3�
2�
�



Tuomet mūsų sudaryta lentelė atrodys taip:

Iš jos galime pasakyti, kad užšifruojant pirmiausia reikia imti antrojo, po to septintojo ir t. t. stulpelio raides.





const MAX = 7; { maksimalus raidžių skaičius rakte }


type lentelė = array [1..MAX] of integer;





procedure sud_lentelę (raktas: string; var num: lentelė);


{ sudaro lentelę, kurios i–asis elementas rodo kurį stulpelį reikės imti i-uoju }


  var pirm, ilgis, i, j: integer;


begin


  ilgis := length (raktas);


  for i := 1 to ilgis do


    begin


      { išrenkame raidę, esančią arčiausiai abėcėlės pradžios }


      pirm := 1;


      while raktas[pirm] = ' ' do


        pirm := pirm + 1;


      for j := 2 to ilgis do


        if (raktas[j] < raktas[pirm]) and 


           (raktas[j] <> ' ')


           then pirm := j;


      { įsimename jos numerį }


      num[i] := pirm;


      { tą raidę pakeičiame tarpu }


      raktas[pirm] := ' ';


    end;


end; { sud_lentelę }





Šifruojant tekstą visai nebūti kurti dvimatę lentelę ir joje grupuoti teksto simbolius. Imkime sąlygoje pateiktą pavyzdį. Sakykime, kad mums reikia antrojo stulpelio raidžių. Rakto ilgis lygus 5. Jei imsime 2–ą, (5+2)–ą, (5+5+2)–ą ir t. t. teksto raides, gausime antrojo stulpelio raides eilės tvarka.


Bendru atveju, stulpelio, kurio numeris stulp raidė r bus lygi pradinio teksto raidei, kurios numeris (r – 1)(st_skaičius + stulp. Čia st_skaičius reiškia bendrą stulpelių skaičių, kuris kaip žinome lygus rakto ilgiui.





procedure šifruoti (raktas, tekstas: string; 


                    var rez: string);


{ užšifruoja duotą tekstą pagal duotą raktą }


  var num: lentelė; { lentelėje surašyta kokia tvarka reikės imti }


                    { stulpelius }


      stulp, stulp_sk, ilgis, i, j: integer;


begin


  sud_lentelę (raktas, num);


  stulp_sk := length (raktas); { stulpelių skaičius }


  ilgis := length (tekstas);   { teksto ilgis }


  rez := '';


  for i := 1 to stulp_sk do


    begin


      stulp := num[i]; { imsime num[i]-ojo stulpelio raides }


      for j := 1 to ilgis div stulp_sk do


        rez := rez + tekstas[(j-1)*stulp_sk + stulp];


      { jei paskutinė eilutė nepilna pridedame raidę iš jos } 


      if ilgis div stulp_sk * stulp_sk + stulp <= ilgis


         then rez := rez +


           tekstas[ilgis div stulp_sk * stulp_sk + stulp]


    end;


end; { šifruoti }








181. Varlių koncertas. Pradžioje tarkime, kad turime tik dvi varles. Viena jų iššoka kas dvi, kita – kas tris minutes. Sakykime, kad pirmą kartą jos iššoko nuliniu laiko momentu.

Iš lentelės matome, kad varlė sukvarksi tais laiko momentais, kurie yra jos kvarksėjimo periodo kartotiniai. T. y. pirmoji varlė kvarksi laiko momentais 2k, antroji – laiko momentais 3k, čia k ( 0, 1, 2, .... Abi varlės kvarksės tuo pat metu tokiais laiko momentais, kurie bus ir pirmos ir antros varlės periodų kartotiniai. 


Toks laiko momentas bus abiejų varlių periodų mažiausias bendras kartotinis mbk bei laiko momentai k(mbk, čia k ( 0, 1, 2, .... 


Pavyzdyje pateiktu atveju kvarksėjimo periodai yra 2 ir 3, o jų mažiausias bendras kartotinis yra 6. Taigi, abiejų varlių koncertai vyks laiko momentais 0, 6, 12, 18, ...


Jei turime daugiau varlių, sprendimas analogiškas: reikia ieškoti visų periodų mažiausio bendro kartotinio.


Dviejų skaičių mažiausio bendro kartotinio ieškosime pagal formulę:


mbk(x, y) ( x div dbd (x, y) ( y


čia, dbd(x, y) yra skaičių x ir y didžiausias bendrasis daliklis. Jį galima rasti pagal Euklido algoritmą. 


Sakykime, kad žinome skaičių p1, p2, ...pk mažiausią bendrą kartotinį kart. Skaičių p1, p2, ...pk, pk+1 mažiausias bendras kartotinis lygus mbk(kart, pk+1).





program varliu_koncertas;


  const MAX = 10; { maksimalus varlių skaičius }


  type periodai = array [1..MAX] of integer;





  function dbd (x, y: longint): longint;


  { randa skaičių x ir y bendrąjį didžiausią daliklį }


  begin


    if x = 0


       then dbd := y


       else dbd := dbd (y mod x, x);


  end; { dbd }





  function mbk (x, y: longint): longint;


  { randa skaičių x ir y bendrąjį mažiausią kartotinį }


  begin


    mbk := x div dbd (x, y) * y


  end; { mbk }





  var val, min, i, kart, varlės: longint;


      p: periodai;


begin


  readln (varlės);


  for i := 1 to  varlės do


    read (p[i]);


  kart := p[1];


  for i := 2 to varlės do


    kart := mbk (kart, p[i]);


  val := kart div 60;


  min := kart mod 60;


  write ('Koncertas įvyks po ', val, ' val. ir ');


  writeln (min, ' min.');


end.





Vyresniųjų grupė


182. Seifo kodas. Ieškomo skaičiaus (seifo kodo) liekana iš 5 (pažymėkime ją liek5) nulemia paskutinįjį skaičiaus skaitmenį. Jis lygus arba tai liekanai (t. y. liek5) arba skaičiui liek5 + 5. Pavyzdžiui, jei skaičiaus liekana iš 5 lygi 3, tai skaičius baigiasi arba skaitmeniu 3 arba skaitmeniu 8. 


Taigi, imame mažiausią n–ženklį skaičių sk (jei n ( 2, tai sk(10, jei n ( 3, tai sk(100 ir t. t.) ir perrenkame n–ženklius skaičius besibaigiančius skaitmenimis liek5 ir liek5+5.


Kyla klausimas ar ilgai reikės ieškoti. Ar nereiks perrinkti apie dvidešimt milijonų skaičių (20000000) jei duotasis skaičius devynženklis?


Atkreipkite dėmesį į vieną įdomią natūraliųjų skaičių savybę:


a mod b ( (a + b(daugiklis) mod b


T. y. jei prie skaičiaus a pridėsime skaičiaus b kartotinį, tai skaičiaus a liekana iš b ir tos sumos liekana iš b sutaps.


Tačiau mus domina liekanos iš trijų skaičių. Mažiausias skaičius kuris be liekanos dalijasi iš 5, 7 ir 11 lygus 5(7(11 ( 385.


Tai reiškia, kad visos įmanomos liekanų kombinacijas galima gauti išanalizavus skaičius iš intervalo [sk, sk+385[. Toliau liekanos ims kartotis. Taigi mažiausias sprendinys būtinai priklauso šiam intervalui. Nesvarbu keliaženklis yra kodas.


Jei n–ženklis skaičius x yra sprendinys, tai visi n–ženkliai skaičiai pavidalo x + 385(daugiklis (daugiklis ( 0, sveikasis skaičius). Taip pat yra sprendiniai. Visų sprendinių skaičių galime užrašyti formule:


(10(sk – 1 –x) div 385 + 1.


const MBK = 5*7*11; { mažiausias bendras kartotinis }





procedure ieškok (n, liek5, liek7, liek11: longint;


                  var kodas, kiek: longint);


  var sk, i, prad: longint;


begin


  { sudarome mažiausią teigiamą n–ženklį skaičių }


  sk := 1;


  for i := 2 to n do


    sk := sk * 10;


  { perrinksime n–ženklius skaičius, kurių liekana iš 5 lygi liek5 }


  prad := sk + liek5;


  { dar neradome skaičiaus tinkamo būti seifo kodu }


  kodas := 0;


  while (prad < sk + MBK) and (prad < sk*10) and 


        (kodas = 0) do


    { jei sk dviženklis, tai sk + MBK jau triženklis skaičius }


  if (prad mod 7 = liek7) and (prad mod 11 = liek11)


     then kodas := prad


     else prad := prad + 5;


  { rasime sprendinių skaičių }


  kiek := (sk * 10 - 1 - kodas) div MBK + 1;


end; { ieškok }








183. Sandaugos įstrižainėse. Uždavinį galime išskaidyti į dvi tarpusavyje mažai susijusias dalis: patikrinimo, ar kvadratas tenkina reikalaujamą sąlygą (t. y. ar jo pagrindinėse įstrižainėse esančių skaičių sandaugos yra lygios) ir visų įmanomų kvadratų perrinkimo.


Pradėkime nuo tikrinimo, ar turimas kvadratas tenkina reikalaujamą sąlygą. Iš karto verta nagrinėti bendrą atvejį: kai tikrinamasis kvadratas yra pradinio kvadrato dalis. 


const MAX = 10; { maksimalus lentelės dydis }


type kvadrat = array [1..MAX, 1..MAX] of longint;





{ žemiau aprašyti globalieji kintamieji, kurių reikšmės }


{ procedūrose nekeičiamos }


var n: longint; { duotojo kvadrato dydis }


    k: kvadrat; { duotasis kvadratas }





function lygios (prad_e, prad_s, dydis: longint):


                                                boolean;


{ tikrina, ar kvadrato, kurio kraštinė lygi kint dydis reikšmei }


{ ir kurio kairiojo viršutinio langelio koordinatės yra (prad_e, prad_s) }


{ įstrižainėse esančių skaičių sandaugos lygios }


  var san_k, san_d, i: longint;


begin


  { skaičių įstrižainėse sandauga }


  san_k := 1; { iš kairės į dešinįjį apatinį kampą }


  san_d := 1; { iš dešinės į kairįjį apatinį kampą }


  for i := 1 to dydis do


    begin


     san_k := san_k * k[prad_e+i-1,  prad_s+i-1];


     san_d := san_d * k[prad_e+i-1, prad_s+(dydis-i+1)-1]


    end;


  lygios := san_k = san_d;


end; { lygios }





Perrinkti visus įmanomus kvadratus nesunku. Pirmiausia ieškome kvadrato, kurio kraštinė lygi n, t. y. patikriname pradinį kvadratą. Jei sąlyga netenkinama, tikriname visus kvadratus, kurių kraštinė lygi n – 1, jei ir vėl neradome – tikriname visus kvadratus, kurių kraštinės lygios n – 2 ir t. t. Blogiausiu atveju kvadrato kraštinė sumažėja iki 1. O bet kuris vienetinis kvadratas tikrai tenkina šią sąlygą. 

Visus kvadratus, kurių kraštinės lygios konkrečiai reikšmei randame keisdami viršutinio kairiojo kampo koordinates. Pavyzdžiui, pradinė lentelė yra tokia:


Sakykime, reikia perrinkti kvadratus, kurių kraštinės lygios 2. Pirmojo kvadrato kairiojo viršutinio kampo koordinatės yra (1, 1). Toliau keičiame kairiojo viršutinio kampo koordinates ((1, 2), (1, 3), (2, 1), (2, 2), (2, 3) ir t. t. ) ir gauname naujus kvadratus:

procedure ieškoti ({ globalieji – n, k }


                   var dydis: longint;


                   var rast_kv: kvadrat { rastasis kvadratas });


  var rasta: boolean;


      paskutinis, i, j, prad_e, prad_s: longint;


begin


  rasta := false; { ieškomo kvadrato dar nerasta }


  dydis := n; { pradėsime nuo didžiausio kvadrato }


  repeat


    prad_e := 1;   { ieškomo kvadrato viršutinis kairysis kampas }


    prad_s := 0;


    paskutinis := n - dydis + 1;


    repeat


      { imame kitą kvadratą }


      if prad_s = paskutinis


         then begin prad_s := 1;


                    prad_e := prad_e + 1


              end


         else prad_s := prad_s + 1;


      { tikriname, ar turimas kvadratas yra sprendinys }


      rasta := lygios (prad_e, prad_s, dydis);


    until rasta or 


       ((prad_e = paskutinis) and (prad_s = paskutinis));


    dydis := dydis - 1;


  until rasta; { vienetinis kvadratas tiks visada, tad visada rasime rezultatą }


  { suformuojame rezultatą }


  dydis := dydis + 1;


  for i := 1 to dydis do


    for j := 1 to dydis do


      rast_kv[i, j] := k[prad_e + (i-1), prad_s + (j-1)];


end; { ieškoti }

184. Kvadratinė Koch sala. Tai fraktalinė kreivė. Ją reikia brėžti remiantis sąlygoje apibrėžtu rekursiniu procesu.


� EMBED Visio.Drawing.5  ���


Prieš aiškinant kaip nubrėžti kreivę, reiktų atkreipti dėmesį į klaidą, kurią labai lengva padaryti.


Sakykime, atkarpą pakeitėme laužte:








Brėžiant aukštesnio laipsnio kreivę reikės kiekvieną laužtės atkarpą vėl keisti laužte. Tada svarbu nepamiršti, kad ilgiausią laužtės atkarpą (piešinyje pažymėtą ***) reikia keisti viena, o ne dviem laužtėmis. 


� EMBED Visio.Drawing.5  ���


Nekorektiškai nubrėžta Koch kvadratinės salos dalis: ilgoji atkarpa pakeista dviem laužtėmis





� EMBED Visio.Drawing.5  ���


Teisingai nubrėžta Koch kvadratinės salos dalis: ilgoji atkarpa pakeista viena laužte





Sąlygoje pateiktuose nurodymuose kaip brėžti kreivę jau sutinkama rekursija. Tad žodžiais aprašytą algoritmą belieka užrašyti programavimo kalba. 


Laisvai pasirenkame kvadrato (pirmojo laipsnio kreivės) kraštinės ilgį. Tuomet kiekvieną kartą atkarpą keičiant laužte, keičiamos atkarpos ilgis dalijamas į keturias dalis. Taip sužinomi laužtės atkarpų ilgiai.


Vienintelė problema, kuri gali iškilti – tai taškų (t. y. laužtės atkarpų galų) koordinačių apskaičiavimas. Paprasčiausias variantas: atkarpą keičiant laužte įsiminti kokia kryptimi (t. y. aukštyn, žemyn, kairėn, dešinėn) brėžiama kiekviena laužtės atkarpa.





� EMBED Visio.Drawing.5  ���


 Išmanantys vektorinę geomet�riją gali tai padaryti dar efektyviau.


Vektoriumi vadinama kryptį turinti atkarpa. Turime bet kokį vektorių � EMBED Equation.2  ��� (žr. brėžinį). Jis išeina iš taško O(0, 0) ir eina į tašką A(x1, y1). Sukonstruosime tokį vektorių � EMBED Equation.2  ���, kad jis būtų to paties ilgio kaip � EMBED Equation.2  ��� ir būtų jam statmenas. Šių vektorių skaliarinė sandauga � EMBED Equation.2  ���, nes jie vienas kitam statmeni. Kadangi skaliarinę sandaugą galime užrašyti per vektorių koordinates, tai turime: x1x2+ y1y2 = 0. Kadangi � EMBED Equation.2  ��� ilgis turi sutapti su � EMBED Equation.2  ��� ilgiu tai � EMBED Equation.2  ���, iš to seka x1x1 + y1y1 = x2x2 + y2y2. Išsprendę šių lygčių sistemą gauname � EMBED Equation.2  ���(x2, y2) = � EMBED Equation.2  ���(-y1, x1) arba � EMBED Equation.2  ���(x2, y2) = � EMBED Equation.2  ���(y1, –x1). Pastebėję, kad abu galimi atsakymai yra priešingi vektoriai, pirmąjį pažymėkime � EMBED Equation.2  ���, o antrąjį (� EMBED Equation.2  ���). Taip pat galima pastebėti, kad jei mes „judėtume“ vektoriaus � EMBED Equation.2  ��� kryptimi, tai � EMBED Equation.2  ���atitinka posūkį į dešinę, o (� EMBED Equation.2  ���) – į kairę. Taip pat svarbu pažymėti, kad „judant“ vektoriais prie einamųjų koordinačių prisideda vektorių koordinatės (žr. paveikslą). 


Lieka tik apskaičiuoti vektoriaus � EMBED Equation.2  ���koordinates (jos programoje vadinamos dx ir dy). Kadangi vektoriaus � EMBED Equation.2  ��� kryptis sutampa su keičiamo vektoriaus (atkarpos) kryptimi, o ilgis 4 kartus mažesnis, tai � EMBED Equation.2  ��� koordinatės yra 4 kartus mažesnės už keičiamo vektoriaus (atkarpos) koordinatėmis.








� EMBED Visio.Drawing.5  ���


Sakykime, vektorių, jungiantį taškus (x1, y1) ir (x2, y2) reikia pakeisti laužte. Kairėje esančiame paveiksle tie taškai bus (0, 1) ir (4, 1). 


Laužtės vektorių koordi�na�tės bus keturis kartus ma�žes�nės nei pradinio vektoriaus. 





Jas ir apskaičiuojame:


dx ( (x2 – x1) div 4;


dy ( (y2 – y1) div 4;


Mūsų atveju dx ( 1, dy ( 0


Pirmoji atkarpa brėžiama nuo taško (x1, y1) iki taško (x1 + dx, y1 + dy). Piešinyje tai atitiks atkarpą nuo jungiančią taškus (0, 1) ir (0, 1) + (1, 0) = (1, 1). Kitame žingsnyje mums reikia tokio pat ilgio vektoriumi pajudėti į kairę (statmenai). Prisiminę, kad šiuo atveju vektoriaus koordinatės turi būti ( dy, -dx) = (0, -1), taigi antras taškas bus (1, 1) + (0, -1) = (1, 0). Trečias gaunamas vėl pajudėjus pradiniu (dx, dy): (1, 0) + (1, 0) = (2, 0). Ketvirtas gaunamas paskus į dešinę ( -dy, dx) ir paėjus per įprastą atstumą 2 kartus: (2, 0) + (0, 1) + (0, 1) = (2, 2). Atitinkamai gaunami ir likę taškai (3, 2), (3, 1), (4, 1).





Pateikiame programą, brėžiančią kvadratinę Koch salą:


program kvadratinė_Koch_sala;


  uses graph; { vartosime modulį graph }





  procedure piešti (n: integer);





   procedure Koch (x1, y1, x2, y2, n: integer);


   { Rekursinė procedūra, piešianti n–ojo laipsnio Koch salos kreivės dalį:


    atkarpa jungianti taškus (x1, y1) ir (x2, y2) pakeičiama laužte }


      var dx, dy: integer;


   begin


     if n = 1


        then line (x1, y1, x2 ,y2)


     else begin


       { atkarpą keičiant laužte, atkarpa padalijama į keturias dalis }


       { apskaičiuojame trumpiausios laužtės atkarpos ilgį }


       dx := (x2 - x1) div 4;


       dy := (y2 - y1) div 4;


       { atkarpą keičiame laužte iš septynių atkarpų } 


       Koch (x1,         y1,         


             x1+dx,      y1+dy,      n-1);


       Koch (x1+dx,      y1+dy,      


             x1+dx+dy,   y1+dy-dx,   n-1);


       Koch (x1+dx+dy,   y1+dy-dx,   


             x1+2*dx+dy, y1+2*dy-dx, n-1);


       Koch (x1+2*dx+dy, y1+2*dy-dx, 


             x1+2*dx-dy, y1+2*dy+dx, n-1);


       Koch (x1+2*dx-dy, y1+2*dy+dx, 


             x1+3*dx-dy, y1+3*dy+dx, n-1);


       Koch (x1+3*dx-dy, y1+3*dy+dx, 


             x1+3*dx,    y1+3*dy,    n-1);


       Koch (x1+3*dx,    y1+3*dy,    


             x2,         y2,         n-1);


     end;


   end; { Koch }





    var cx, cy: integer;


  begin


    cx := GetMaxX div 2; { gauname ekrano centro koordinates }


    cy := GetMaxY div 2;


    { brėžiame kiekvieną iš keturių kreivės dalių }


    Koch (cx-cy div 2, cy div 2,      


          cx+cy div 2, cy div 2,      n);


    Koch (cx+cy div 2, cy div 2,      


          cx+cy div 2, cy + cy div 2, n);


    Koch (cx+cy div 2, cy + cy div 2, 


          cx-cy div 2, cy + cy div 2, n);


    Koch (cx-cy div 2, cy + cy div 2, 


          cx-cy div 2, cy div 2,      n);


  end; { piešti }





  var n,                { kreivės laipsnis }


      tvarkyklė,        { grafinės plokštės tvarkyklė }


      režimas: integer; { grafinis režimas }


begin


  readln (n); { įvedamas kreivės laipsnis }


  { automatiškai nustatoma grafinės plokštės tvarkyklė }


  tvarkyklė := detect;


  { inicializuojamas grafinis režimas }


  InitGraph (tvarkyklė, režimas,''); 


  piešti (n); { piešiama }


  readln;


  CloseGraph;  { grįžtama į tekstinį režimą }


end.

